

SAP Hybris Marketing Cloud 1805

Document Version: 1.0 – 2018-05-07 –


Enable Navigation from SAP Customer Relationship Management to SAP Hybris Marketing Cloud

Contents

1	Introduction	3
2	Prerequisites	4
3	Create New Classes	5
4	Maintain HTTP-Destination for SAP Hybris Marketing Cloud.....	8
4.1	Assign the HTTP-Destination to the Logical System	8
5	Integrate SAP Hybris Marketing Cloud Factsheet	9
5.1	Create Button to Open Factsheet	9

1 Introduction

This guide is intended for customers and partners dealing with the integration of SAP Hybris Marketing Cloud with SAP Customer Relationship Management (SAP CRM). This guide describes how to integrate the SAP Hybris Marketing Cloud factsheets for *Corporate Accounts*, *Contacts* and *Consumers* into the SAP Customer Relationship Management WebClient UI.



The URLs can be customized separately for development, test and productive clients.
At runtime, the complete URL is provided by the implemented classes.

The SAP Hybris Marketing Cloud URL of the factsheet looks as follows:

Contacts, Consumers and Accounts:

`https://<server>:<port>/sap/bc/ui5_ui2/ushell/shells/abap/FioriLaunchpad.html#MarketingContact-displayFactSheet?OriginID=SAP_CRM_BUPA&InteractionContactID=<CRM_CONTACT_ID>`

2 Prerequisites

This guide uses the SAP CRM Enhancement concept to show the SAP Hybris Marketing Cloud factsheet buttons. As prerequisite, an active *Enhancement Set* must be created, and assigned to the actual client.

If your system does not have an active *Enhancement Set* assigned yet follow this procedure:

Create Enhancement Set


- Open transaction **SM30**, enter as table *BSPWDV_EHSET_DEF* and choose *Maintain*
- In case you do not have an active Enhancement Set yet, choose *New Entries*
- Provide a key for your Enhancement Set, and a description
- Choose *Save*.

Assign Enhancement Set to your client

- Open transaction **SM30**, enter as table *BSPWDV_EHSET_ASG* and choose *Maintain*.
- In case you do not have the Enhancement Set assigned to your actual client yet proceed as follows:
 - Choose *New Entries* to create a new entry.
 - Provide an *Assignment No.* as key for the entry (any number not yet used).
 - Enter the actual client of your system.
 - Use F4 to choose the prior defined *Enhancement Set*.

3 Create New Classes

- Open transaction **SE24**, and create a new class, **YMKT_URL_DESTINATION**.
The class is used to determine the SAP Hybris Marketing Cloud destination.
- Navigate to the *Source Code Based* view.



- Delete the displayed source code, and enter the following code:

```
class YMKT_URL_DESTINATION definition
    public
    create public .

public section.

    class-methods GET_URL_PREFIX
        returning
            value(RV_VALUE) type STRING .
protected section.
*** protected components of class YMKT_URL_DESTINATION
*** do not include other source files here!!!
private section.
*** private components of class YMKT_URL_DESTINATION
*** do not include other source files here!!!
ENDCLASS.

CLASS YMKT_URL_DESTINATION IMPLEMENTATION.

* <SIGNATURE>-----+
* | Static Public Method YMKT_URL_DESTINATION=>GET_URL_PREFIX
* +-----+
* | [ <- () ] RV_VALUE                                TYPE          STRING
* +-----+</SIGNATURE>

METHOD get_url_prefix.

    DATA lr_value                TYPE REF TO string.
    DATA lv_own_logical_system    TYPE tbdls-logsys.
    DATA lv_rfc_destination       TYPE rfcddest.
    DATA lv_target_port           TYPE rfcdisplay-rfcsysid.
    DATA lv_target_host           TYPE rfcdisplay-rfchost.
    DATA lv_ssl                   TYPE rfcdisplay-rfcsnc.
    DATA lv_protocol              TYPE string.
    DATA lv_url                   TYPE string.
    DATA lv_url_part              TYPE string.
    DATA ls_build_header          TYPE crmst_header_object_build.
    DATA lv_partner               TYPE bu_partner.
    DATA lv_parname3              TYPE smofparsfa-parname3.

    CREATE DATA lr_value.

    CALL FUNCTION 'OWN_LOGICAL_SYSTEM_GET'
        IMPORTING
            own_logical_system      = lv_own_logical_system
    EXCEPTIONS
        own_logical_system_not_defined = 1
        OTHERS                      = 2.

    IF sy-subrc <> 0.
* Implement suitable error handling here
    ENDIF.

    lv_parname3 = lv_own_logical_system.

    CALL FUNCTION 'SMOF0_READ_SMOFPARSFA'
        EXPORTING
            parsfakey = 'HPA'
            paraname  = 'CEI'
            parname2  = 'RFC_DEST'
            parname3  = lv_parname3
        IMPORTING
            e_parvall = lv_rfc_destination
```

```

        EXCEPTIONS
            OTHERS      = 99.
    IF ( sy-subrc <> 0 ) OR ( lv_rfc_destination IS INITIAL ).
        RETURN.
    ENDIF.

    CALL FUNCTION 'RFC_READ_HTTP_DESTINATION'
    EXPORTING
        destination      = lv_rfc_destination
    IMPORTING
        servicenr        = lv_target_port
        server           = lv_target_host
        ssl              = lv_ssl
    EXCEPTIONS
        authority_not_available = 1
        destination_not_exist   = 2
        information_failure     = 3
        internal_failure        = 4
        no_http_destination     = 5
        OTHERS                  = 6.

    IF sy-subrc <> 0.
        RETURN.
    ENDIF.

    IF NOT lv_ssl IS INITIAL.
        lv_protocol = 'https'.
    ELSE.
        lv_protocol = 'http'.
    ENDIF.

    IF NOT lv_target_port IS INITIAL.
        CONCATENATE lv_protocol '://' lv_target_host ':' lv_target_port INTO rv_value.
    ELSE.
        CONCATENATE lv_protocol '://' lv_target_host INTO rv_value.
    ENDIF.

    ENDMETHOD.
ENDCLASS.

```

1

Note

The code contains a reference to the Middleware Parameter you will define in section 4.1. So, if you choose other Middleware Parameter key and parameter names you have to adapt the code accordingly.

- Create a new class, YMKT_URL_DEST_MASH_PARTNER.
The class is used to determine the URL for the factsheet mashup.
- Navigate to the *Source Code Based* view.
- Delete the displayed source code, and enter the following code:

```

class YMKT_URL_DEST_MASH_PARTNER definition
public
    inheriting from YMKT_URL_DESTINATION
    final
    create public .

```

public section.

```

*** public components of class YMKT_URL_DEST_MASH_PARTNER
*** do not include other source files here!!!
    interfaces IF_WCF_CDS_CUSTOM_FUNCTION .
protected section.
*** protected components of class YMKT_URL_DEST_MASH_PARTNER
*** do not include other source files here!!!
private section.
*** private components of class YMKT_URL_DEST_MASH_PARTNER
*** do not include other source files here!!!
ENDCLASS.

```

```

CLASS YMKT_URL_DEST_MASH_PARTNER IMPLEMENTATION.

```

```

* <SIGNATURE>-----+
* | Static Public Method YMKT_URL_DEST_MASH_PARTNER=>IF_WCF_CDS_CUSTOM_FUNCTION~GET_VALUE

```

```

* +-----+
* | [--->] IR_REFERENCE_ENTITY          TYPE REF TO CL_CRM_BOL_ENTITY
* | [<-( )] RR_VALUE                   TYPE REF TO DATA
* +-----+</SIGNATURE>
METHOD if_wcf_cds_custom_function~get_value.

    DATA lr_value          TYPE REF TO string.
    DATA lv_url            TYPE string.
    DATA lv_url_part       TYPE string.
    DATA ls_buil_header    TYPE crmst_header_object_buil.
    DATA lv_partner        TYPE bu_partner.
    DATA lv_prefix TYPE string.

    CREATE DATA lr_value.

    CALL METHOD get_url_prefix
      RECEIVING
        rv_value = lv_prefix.

    ir_reference_entity->get_properties( IMPORTING es_attributes = ls_buil_header ).

    CASE ls_buil_header-bp_category.
      WHEN '1' OR '2'. " Person or Organisation

        lv_url_part = '/sap/bc/ui5_ui5/ui2/ushell/shells/abap/FioriLaunchpad.html#MarketingContact-
displayFactSheet?OriginID=SAP_CRM_BUPA'.
      WHEN OTHERS.
        EXIT. "Group not supported
    ENDCASE.

    lv_partner = ls_buil_header-bp_number.

    CONCATENATE lv_prefix lv_url_part '&InteractionContactID=' lv_partner INTO lv_url.

    lr_value->* = lv_url.
    rr_value = lr_value.

ENDMETHOD.
ENDCLASS.

```

4 Maintain HTTP-Destination for SAP Hybris Marketing Cloud

- Open transaction **SM59** and maintain the following attributes:
 - Target Host
 - (optional) Port (Service No.)
 - Flag **SSL Active** for a **HTTPS** Protocol

The image displays two screenshots of the SAP SM59 transaction. The left screenshot shows the 'RFC Destination CC2CLNT350_HTTP' configuration. The 'Target System Settings' section has 'Target Host' set to 'ccf-350.wdf.sap.corp' and 'Service No.' highlighted. The 'HTTP Proxy Options' section is also visible. The right screenshot shows the 'Logon & Security' tab. The 'Status of Secure Protocol' section has 'SSL' set to 'Active' and 'SSL Certificate' set to 'DEFAULT SSL Client (Standard)'. The 'Logon Procedure' section shows 'Do Not Use a User' selected.

4.1 Assign the HTTP-Destination to the Logical System

- Open transaction **R3AC6**.
- Maintain the following parameters:
 - Key: HPA
 - Parameter1: CEI
 - Parameter2: RFC_DEST
 - Parameter3: own <logical system>
 - Value: prior defined HTTP-Connection

- Maintain also the RFC-Destinations for your test and productive systems, and transport them.



Note

The logical systems (Parameter3) of the test and productive Systems are different.



5 Integrate SAP Hybris Marketing Cloud Factsheet



5.1 Create Button to Open Factsheet

Note

The following steps require knowledge of the Enhancement Concept of SAP CRM.

For more information,


see https://help.sap.com/viewer/p/SAP_CUSTOMER_RELATIONSHIP_MANAGEMENT  Application Help 

SAP Library  WebClient UI Framework  Framework Enhancement


(<https://help.sap.com/viewer/62055d1ef81d43f3aa41473061d86046/7.0.4.10/en-US/2da4278984b84ec492ff89e1c62c879f.html>)

To create a button to open factsheet, proceed as follows:

- Open CRM UI Component Workbench via transaction **BSP_WD_CMPWB**.
- Provide **BP_HEAD** as component and enter your *Enhancement Set* (use F4 for a list).
If the Enhancement Set is assigned to your actual client it is prefilled when you open the transaction proceed as follows:



- Choose *Display* to open the component
- If the component is not yet enhanced choose the button *Enhance Component*.
(otherwise the button to delete the component enhancement is shown)



- In the following popup provide the created *Enhancement Set* (use F4 for help).
- In the following popup provide the Application **ZBP_HEAD**.

Note


All component-related generated content is grouped underneath the provided *Application*.

We recommended to use a similar name for the *Application* as the one of the enhanced component.


(In this case ZBP_HEAD, but you can enter any name)

Technically, the name is used to create a BSP-Application.


- Confirm the popup by Yes.




- Confirm the popup.




- Choose a package in customer namespace (Z or Y) for your development.
If you do not have a package yet, create the package in transaction **SE80**.
- Enhance view **BP_HEAD/BPHEADOverview** by choosing the view, and make use of the context menu (with right-click) – choose **Enhance**.



- Confirm all popups for the generated objects.
If you get the message *Class <CLASS> already exists*, enter a different name for the generated class, and confirm the popup.
- Choose the generated Z-Class by double-clicking the class with the ending **.IMPL**.



- Switch to change mode (1).



- Make sure the filter for inherited methods is not set (2).
- Redefine Method `IF_BSP_WD_TOOLBAR_CALLBACK~GET_BUTTONS` of the generated `.IMPL` class by positioning the cursor to the method, and clicking the button for redefinition (3) and enter following code.

```
METHOD if_bsp_wd_toolbar_callback~get_buttons.

    DATA lr_entity TYPE REF TO cl_crm_bol_entity.
    DATA lr_link   TYPE REF TO string.
    DATA lv_link   TYPE string.
    DATA ls_button TYPE crmt_thtmlb_button_ext.


    CALL METHOD super->if_bsp_wd_toolbar_callback~get_buttons
        RECEIVING
            rt_buttons = rt_buttons.

    lr_entity ?= me->if_bsp_wd_history_state_descr~get_main_entity( ).

    IF lr_entity IS BOUND.

        lr_link ?= ymkt_url_dest_mash_partner=>if_wcf_cds_custom_function~get_value( lr_entity ).
        lv_link = lr_link->*.

        ls_button-on_client_click = 'javascript:window.open( "' && lv_link && '" )'.
        ls_button-id              = 'YMKT_FACTSHEET'.
        ls_button-page_id         = me->component_id.
        ls_button-enabled         = abap_true.
        ls_button-text            = 'yMKT-Factsheet'.
        APPEND ls_button TO rt_buttons.
        CLEAR ls_button.
    ENDIF.
ENDMETHOD.
```



© 2018 SAP SE or an SAP affiliate company. All rights reserved.

No part of this publication may be reproduced or transmitted in any form or for any purpose without the express permission of SAP SE or an SAP affiliate company. The information contained herein may be changed without prior notice.

Some software products marketed by SAP SE and its distributors contain proprietary software components of other software vendors. National product specifications may vary.

These materials are provided by SAP SE or an SAP affiliate company for informational purposes only, without representation or warranty of any kind, and SAP or its affiliated companies shall not be liable for errors or omissions with respect to the materials. The only warranties for SAP or SAP affiliate company products and services are those that are set forth in the express warranty statements accompanying such products and services, if any. Nothing herein should be construed as constituting an additional warranty.

SAP and other SAP products and services mentioned herein as well as their respective logos are trademarks or registered trademarks of SAP SE (or an SAP affiliate company) in Germany and other countries. All other product and service names mentioned are the trademarks of their respective companies.

Please see <http://www.sap.com/corporate-en/legal/copyright/index.epx> for additional trademark information and notices.